# Maintainer: Felix Yan <felixonmars@archlinux.org>

_hkgname=hls-tactics-plugin
pkgname=haskell-hls-tactics-plugin
pkgver=1.8.0.0
pkgrel=101
pkgdesc="Wingman plugin for Haskell Language Server"
url="https://haskellwingman.dev"
license=("Apache")
arch=('loong64' 'x86_64')
depends=('ghc-libs' 'haskell-aeson' 'haskell-extra' 'haskell-fingertree' 'haskell-generic-lens'
         'haskell-ghc' 'haskell-ghc-exactprint' 'haskell-ghc-source-gen' 'haskell-ghcide'
         'haskell-hls-graph' 'haskell-hls-plugin-api' 'haskell-hls-refactor-plugin'
         'haskell-hyphenation' 'haskell-lens' 'haskell-lsp' 'haskell-megaparsec'
         'haskell-parser-combinators' 'haskell-prettyprinter' 'haskell-refinery' 'haskell-retrie'
         'haskell-syb' 'haskell-unagi-chan' 'haskell-unordered-containers')
makedepends=('ghc' 'uusi' 'haskell-quickcheck' 'haskell-hls-test-utils' 'haskell-hspec'
             'haskell-hspec-discover' 'haskell-hspec-expectations' 'haskell-lsp-types'
             'haskell-tasty-hunit' 'haskell-tasty-hspec')
source=("https://hackage.haskell.org/packages/archive/$_hkgname/$pkgver/$_hkgname-$pkgver.tar.gz")
sha512sums=('f94e3263891b9820adb5d8e8c5d92a1da9e8a0f78bec9d37fd8fd843b62bc55ffc07bbf7cc84f6f9b28a842dcb78d1924518b1fb1d805b314f7b2f63b632a0c0')

prepare() {
  cd $_hkgname-$pkgver
  gen-setup
}

build() {
  cd $_hkgname-$pkgver

  # disable tests because the source tarball has missing files
  runhaskell Setup configure -O --enable-shared --enable-executable-dynamic --disable-library-vanilla \
    --prefix=/usr --docdir=/usr/share/doc/$pkgname --datasubdir=$pkgname --disable-tests \
    --dynlibdir=/usr/lib --libsubdir=\$compiler/site-local/\$pkgid --ghc-option=-fllvm \
    --ghc-option=-optl-Wl\,-z\,relro\,-z\,now \
    --ghc-option='-pie'

  runhaskell Setup build $MAKEFLAGS
  runhaskell Setup register --gen-script
  runhaskell Setup unregister --gen-script
  sed -i -r -e "s|ghc-pkg.*update[^ ]* |&'--force' |" register.sh
  sed -i -r -e "s|ghc-pkg.*unregister[^ ]* |&'--force' |" unregister.sh
}

check() {
  cd $_hkgname-$pkgver
  # TODO
  runhaskell Setup test --show-details=direct || echo "Tests failed"
}

package() {
  cd $_hkgname-$pkgver

  install -D -m744 register.sh "$pkgdir"/usr/share/haskell/register/$pkgname.sh
  install -D -m744 unregister.sh "$pkgdir"/usr/share/haskell/unregister/$pkgname.sh
  runhaskell Setup copy --destdir="$pkgdir"
  rm -f "$pkgdir"/usr/share/doc/$pkgname/LICENSE
}
